Lesson no.36

Storage Access using DOS

BIOS provides raw access to the storage medium while DOS gives a more logical view and more cooked services. Everything is a file. A directory is a specially organized file that is interpreted by the operating system itself. A list of important DOS services for file manipulation is given below.

INT 21 - CREATE OR TRUNCATE FILE

AH = 3Ch

CX = file attributes
DS:DX -> ASCIZ filename

Return: 
CF = error flag
AX = file handle or error code
INT 21 - OPEN EXISTING FILE

AH = 3Dh

AL = access and sharing modes 

DS:DX -> ASCIZ filename

CL = attribute mask of files to look for (server call only)

Return: 
CF = error flag
AX = file handle or error code
INT 21 - CLOSE FILE

AH = 3Eh

BX = file handle

Return: 
CF = error flag

AX = error code 

INT 21 - READ FROM FILE 

AH = 3Fh

BX = file handle

CX = number of bytes to read

DS:DX -> buffer for data

Return: 
CF = error flag 
AX = number of bytes actually read or error code

INT 21 - WRITE TO FILE 

AH = 40h

BX = file handle

CX = number of bytes to write

DS:DX -> data to write

Return: 
CF = error flag

AX = number of bytes actually written or error code

INT 21 - DELETE FILE

AH = 41h

DS:DX -> ASCIZ filename (no wildcards, but see notes)

Return: 
CF = error flag
AX = error code

INT 21 - SET CURRENT FILE POSITION

AH = 42h

AL = origin of move

BX = file handle

CX:DX = offset from origin of new file position

Return: 
CF = error flag
DX:AX = new file position in bytes from start of file

AX = error code in case of error
INT 21 - GET FILE ATTRIBUTES

AX = 4300h

DS:DX -> ASCIZ filename

Return: 
CF = error flag
CX = file attributes 

AX = error code
INT 21 - SET FILE ATTRIBUTES

AX = 4301h

CX = new file attributes 

DS:DX -> ASCIZ filename

Return: 
CF = error flag
AX = error code 

We will use some of these services to find that two files are same in contents or different. We will read the file names from the command prompt. The command string is passed to the program in the program segment prefix located at offset 0 in the current segment. The area from 0-7F contains information for DOS, while the command tail length is stored at 80. From 81 to FF, the actual command tail is stored terminated by a CR (Carriage Retrun).

	; file comparison using dos services
[org 0x0100]

              jmp  start

filename1:    times 128 db 0          ; space for first filename
filename2:    times 128 db 0          ; space for second filename
handle1:      dw   0                  ; handle for first file
handle2:      dw   0                  ; handle for second file
buffer1:      times 4096 db 0         ; buffer for first file
buffer2:      times 4096 db 0         ; buffer for second file
format:       db   'Usage error: diff <filename1> <filename2>$'

openfailed:   db   'First file could not be opened$'

openfailed2:  db   'Second file could not be opened$'

readfailed:   db   'First file could not be read$'

readfailed2:  db   'Second file could not be read$'

different:    db   'Files are different$'

same:         db   'Files are same$'

start:        mov  ch, 0              
              mov  cl, [0x80]         ; command tail length in cx
              dec  cx                 ; leave the first space
              mov  di, 0x82           ; start of command tail in di
              mov  al, 0x20           ; space for parameter separation
              cld                     ; auto increment mode
              repne scasb             ; search space
              je   param2             ; if found, proceed
              mov  dx, format         ; else, select error message
              jmp  error              ; proceed to error printing
param2:       push cx                 ; save original cx
              mov  si, 0x82           ; set si to start of param
              mov  cx, di             ; set di to end of param
              sub  cx, 0x82           ; find param size in cx
              dec  cx                 ; excluding the space
              mov  di, filename1      ; set di to space for filename 1
              rep  movsb              ; copy filename there
              mov  byte [di], 0       ; terminate filename with 0
              pop  cx                 ; restore original cx 
              inc  si                 ; go to start of next filename
              mov  di, filename2      ; set di to space for filename 2
              rep  movsb              ; copy filename there
              mov  byte [di], 0       ; terminate filename with 0
              mov  ah, 0x3d           ; service 3d – open file
              mov  al, 0              ; readonly mode
              mov  dx, filename1      ; address of filename
              int  0x21               ; dos services
              jnc  open2              ; if no error, proceed
              mov  dx, openfailed     ; else, select error message
              jmp  error              ; proceed to error printing
open2:        mov [handle1], ax       ; save handle for first file
              mov  ah, 0x3d           ; service 3d – open file
              mov  al, 0              ; readonly mode
              mov  dx, filename2      ; address of filename
              int  0x21               ; dos services
              jnc  store2             ; if no error, proceed
              mov  dx, openfailed2    ; else, select error message
              jmp  error              ; proceed to error printing
store2:       mov  [handle2], ax      ; save handle for second file
readloop:     mov  ah, 0x3f           ; service 3f – read file
              mov  bx, [handle1]      ; handle for file to read
              mov  cx, 4096           ; number of bytes to read
              mov  dx, buffer1        ; buffer to read in
              int  0x21               ; dos services
              jnc  read2              ; if no error, proceed
              mov  dx, readfailed     ; else, select error message
              jmp  error              ; proceed to error printing
read2:        push ax                 ; save number of bytes read
              mov  ah, 0x3f           ; service 3f – read file
              mov  bx, [handle2]      ; handle for file to read
              mov  cx, 4096           ; number of bytes to read
              mov  dx, buffer2        ; buffer to read in
              int  0x21               ; dos services
              jnc  check              ; if no error, proceed
              mov  dx, readfailed2    ; else, select error message
              jmp  error              ; proceed to error printing
check:        pop  cx                 ; number of bytes read of file 1 
              cmp  ax, cx             ; are number of byte same
              je   check2             ; yes, proceed to compare them
              mov  dx, different      ; no, files are different
              jmp  error              ; proceed to message printing
check2:       test ax, ax             ; are zero bytes read
              jnz  compare            ; no, compare them
              mov  dx, same           ; yes, files are same
              jmp  error              ; proceed to message printing
compare:      mov  si, buffer1        ; point si to file 1 buffer
              mov  di, buffer2        ; point di to file 2 buffer 
              repe cmpsb              ; compare the two buffers
              je   check3             ; if equal, proceed
              mov  dx, different      ; else, files are different
              jmp  error              ; proceed to message printing
check3:       cmp  ax, 4096           ; were 4096 bytes read
              je   readloop           ; yes, try to read more
              mov  dx, same           ; no, files are same
error:        mov  ah, 9              ; service 9 – output message
              int  0x21               ; dos services
              mov  ah, 0x3e           ; service 3e – close file
              mov  bx, [handle1]      ; handle of file to close
              int  0x21               ; dos services
              mov  ah, 0x3e           ; service 3e – close file
              mov  bx, [handle2]      ; handle of file to close
              int  0x21               ; dos services
              mov  ax, 0x4c00         ; terminate program
              int  0x21


Another interesting service that DOS provides regarding files is executing them. An important point to understand here is that whenever a program is executed in DOS all available memory is allocated to it. No memory is available to execute any new programs. Therefore memory must be freed using explicit calls to DOS for this purpose before a program is executed. Important services in this regard are listed below.

INT 21 - ALLOCATE MEMORY

AH = 48h

BX = number of paragraphs to allocate

Return: 
CF = error flag
AX = segment of allocated block or error code in case of error
BX = size of largest available block in case of error
INT 21 - FREE MEMORY

AH = 49h

ES = segment of block to free

Return: 
CF = error flag
AX = error code

INT 21 - RESIZE MEMORY BLOCK

AH = 4Ah

BX = new size in paragraphs

ES = segment of block to resize

Return: 
CF = error flag

AX = error code

BX = maximum paragraphs available for specified memory block

INT 21 - LOAD AND/OR EXECUTE PROGRAM

AH = 4Bh

AL = type of load (0 = load and execute)
DS:DX -> ASCIZ program name (must include extension)

ES:BX -> parameter block 

Return: 
CF = error flag
AX = error code
The format of parameter block is as follows.

Offset
Size
Description

 00h
WORD
segment of environment to copy for child process 
(copy caller's environment if 0000h)

 02h
DWORD
pointer to command tail to be copied into child's PSP

 06h
DWORD
pointer to first FCB to be copied into child's PSP

 0Ah
DWORD
pointer to second FCB to be copied into child's PSP

 0Eh
DWORD
(AL=01h) will hold subprogram's initial SS:SP on return

 12h
DWORD
(AL=01h) will hold entry point (CS:IP) on return

As an example we will use the multitasking kernel client from the multitasking chapter and modify it such that after running all three threads it executes a new instance of the command prompt instead of indefinitely hanging around.

	; another multitasking TSR caller
[org 0x0100]

              jmp  start

; parameter block layout:

; cs,ip,ds,es,param

;  0, 2, 4, 6,  8
paramblock:   times 5 dw 0            ; space for parameters 
lineno:       dw   0                  ; line number for next thread
chars:        db '\|/-'               ; chracters for rotating bar
message:      db 'moving hello'       ; moving string
message2:     db '            '       ; to erase previous string
messagelen:   dw 12                   ; length of above strings

tail:         db ' ',13

command:      db 'COMMAND.COM', 0

execblock:    times 11 dw 0
;;;;; COPY LINES 028-071 FROM EXAMPLE 10.1 (printnum) ;;;;;
;;;;; COPY LINES 073-114 FROM EXAMPLE 10.1 (printstr) ;;;;;
;;;;; COPY LINES 103-126 FROM EXAMPLE 11.5 (mytask) ;;;;;
;;;;; COPY LINES 128-146 FROM EXAMPLE 11.5 (mytask2) ;;;;;
;;;;; COPY LINES 148-193 FROM EXAMPLE 11.5 (mytask3) ;;;;;
start:        mov  [paramblock+0], cs ; code segment parameter
              mov  word [paramblock+2], mytask ; offset parameter
              mov  [paramblock+4], ds ; data segment parameter
              mov  [paramblock+6], es ; extra segment parameter
              mov  word [paramblock+8], 0 ; parameter for thread
              mov  si, paramblock     ; address of param block in si
              int  0x80               ; multitasking kernel interrupt
              mov  [paramblock+0], cs ; code segment parameter
              mov  word [paramblock+2], mytask2 ; offset parameter
              mov  [paramblock+4], ds ; data segment parameter
              mov  [paramblock+6], es ; extra segment parameter
              mov  word [paramblock+8], 0 ; parameter for thread
              mov  si, paramblock     ; address of param block in si
              int  0x80               ; multitasking kernel interrupt
              mov  [paramblock+0], cs ; code segment parameter
              mov  word [paramblock+2], mytask3 ; offset parameter
              mov  [paramblock+4], ds ; data segment parameter
              mov  [paramblock+6], es ; extra segment parameter
              mov  word [paramblock+8], 0 ; parameter for thread
              mov  si, paramblock     ; address of param block in si
              int  0x80               ; multitasking kernel interrupt
              mov  ah, 0x4a           ; service 4a – resize memory
              mov  bx, end            ; end of memory retained
              add  bx, 15             ; rounding up
              mov  cl, 4

              shr  bx, cl             ; converting into paras
              int  0x21               ; dos services
              mov  ah, 0x4b           ; service 4b - exec
              mov  al, 0              ; load and execute
              mov  dx, command        ; command to be executed
              mov  bx, execblock      ; address of execblock
              mov  word [bx+2], tail  ; offset of command tail
              mov  [bx+4], ds         ; segment of command tail
              int  0x21               ; dos services
              jmp $                   ; loop infinitely if returned
end:


